ECE 2101L
Electrical Circuit Analysis II Laboratory

Lab 7
Input and Output Impedances of AC Black Boxes

Report

Choi Tim Antony Yung

April 5, 2020

Objective

The purpose of this experiment is to study the method of determining input and output impedances of AC black boxes from measurements.

1 Measuring input impedance of a black box





```
ECE2101L Lab7 Part B2: Measuring INPUT impedance of a black box
* Author: Choi Tim Antony Yung

V1 1 0 AC 5 0
* AC voltage source with magnitude of 5V and phase of 0
* with node 1 as + terminal and node 0 (ground) at - terminal

R0 1 2 130
* Resistor 0 from node 1 to node 2 with 130 ohm

R1 2 3 51
* Resistor 1 from node 2 to node 3 with 51 ohm

R2 3 0 620
* Resistor 2 from node 3 to node 0 (ground) with 620 ohm

C0 3 0 33u
* Capacitor 0 from node 3 to node 0 (ground) with 33 microFarad
* u denotes micro

.AC LIN 1 60 60
* perform AC analysis with one frequency value (from 60 Hz to 60 Hz)

.PRINT AC v(2)
* Print the voltage at node 2, relative to ground

.PRINT AC i(R0)
* Print the current flowing through R0

.END
```

```
ECE2101L Lab7 Part B2: Measuring INPUT impedance of a black box
       --- AC Analysis ---

frequency:    60              Hz
V(2):         mag:    2.41621 phase:   -29.7742°        voltage
I(R0):        mag:  0.0241612 phase:    22.4577°        device_current
```

Figure 1: Screenshot of circuit description used and the simulation result

Procedure

A SPICE netlist was written to simulate the above circuit with LTspice XVI, the netlist is attached at the back of this report. The current of R_0 was used to determine I.

Result

V_2 calculated	I calculated	Z_{in} calculated
$2.41621\ \angle -29.7742°$ V	$0.0241612\ \angle 22.4577°$ A	$100.0037\ \angle -52.2319°\ \Omega$

V_2 measured	I measured	Z_{in} measured	$\lvert Z_{in}\rvert$ Error
$2.41621\ \angle -29.7742°$ V	$0.0241612\ \angle 22.4577°$ A	$100.0037\ \angle -52.2319°\ \Omega$	0 %

Analysis

As the above measurement was simulated, there is no error. However, were it to be measured from an actual circuit, the measurement would be subjected to errors due to variation of impedance from its nominal value, imprecise oscilloscope measurement, dissipation of energy from nonideal wire, and noise from electromagnetic interferences, to name a few. Possible improvements to the above problem includes measuring impedances with LCR meter and adjusting impedance by adding small impedances, avoiding long wires, and spacing apart components, et cetera.

2 Measuring output impedance of a black box



Theory

With V_{th} as the open circuit voltage difference of A and B, and V_{01} and V_{02} as the voltage of $R = 130\,\Omega$ and $R = 130\,\Omega$ respectively, the magnitude of output impedance R_{out} and X_{out} can be found by solving the following system of equations:

$$|V_{01}| = \frac{R_1}{\sqrt{(R_1 + R_{out})^2 + (X_{out})^2}} |V_{th}|$$

$$|V_{02}| = \frac{R_2}{\sqrt{(R_2 + R_{out})^2 + (X_{out})^2}} |V_{th}|$$

... which can be simplified to the following:

$$(R_1 + R_{out})^2 + (X_{out})^2 = \left(\frac{R_1}{|V_{01}|} |V_{th}| \right)^2$$

$$(R_2 + R_{out})^2 + (X_{out})^2 = \left(\frac{R_2}{|V_{02}|} |V_{th}| \right)^2$$

By solving the above system of equation above, we can determine the magnitude of R_{out} and X_{out}.

Procedure

A SPICE netlist was written for the blackbox subcircuit, and three separate netlists was written to simulate, in LTspice XVI, the above circuit with A to B open or with different value of resistance attached to A and B, the netlists are attached at the back of this report.



Figure 2: Screenshot of circuit description used and the simulation result leaving A and B open



Figure 3: Screenshot of circuit description used and the simulation result with 130 Ω resistor attached



Figure 4: Screenshot of circuit description used and the simulation result with 620 Ω resistor attached

Result

R	$\lvert V_{th}\rvert$ measured	$\lvert V_0\rvert$ measured	$\lvert R_{out}\rvert$ calculated	$\lvert X_{out}\rvert$ calculated
130 Ω	6 V	4.503 49 V	43 Ω	8.3 Ω
620 Ω	6 V	5.610 42 V		

Analysis

After obtaining the value of the output impedance, we can repeat the measurement with an inductor L attached in series before R. In theory, if X_{out} is inductive and therefore positive, then adding L in series will increase the magnitude of total reactance. Whereas if X_{out} is capacitive therefore negative, then adding L, a positive reactance, in series will decrease the magnitude of total reactance. Therefore, if the measurement of $|X_{out}|$ after adding L is larger than before, then X_{out} is inductive and therefore positive, otherwise, it is capacitive and therefore negative.

```
ECE2101L Lab7 Part B2: Measuring INPUT impedance of a black box
* Author: Choi Tim Antony Yung

V1 1 0 AC 5 0
* AC voltage source with magnitude of 5V and phase of 0
* with node 1 as + terminal and node 0 (ground) at - terminal

R0 1 2 130
* Resistor 0 from node 1 to node 2 with 130 ohm

R1 2 3 51
* Resistor 1 from node 2 to node 3 with 51 ohm

R2 3 0 620
* Resistor 2 from node 3 to node 0 (ground) with 620 ohm

C0 3 0 33u
* Capacitor 0 from node 3 to node 0 (ground) with 33 microFarad
* u denotes micro

.AC LIN 1 60 60
* perform AC analysis with one frequency value (from 60 Hz to 60 Hz)

.PRINT AC v(2)
* Print the voltage at node 2, relative to ground

.PRINT AC i(R0)
* Print the current flowing through R0

.END
```

```
* ECE2101L Lab7 Part B2: Measuring OUTPUT impedance of a black box
* Author: Choi Tim Antony Yung
* Definition of blackbox subcircuit

.SUBCKT BOX A B
V1 1 B AC 6 0
Ra 1 2 43
C1 2 B 10p
C2 2 3 1000u
Rb 3 4 51
C3 4 B 20p
C4 3 A 470u
.ENDS
```

```
ECE2101L Lab7 Part B2: Measuring OUTPUT impedance of a black box with A to B open
* Author: Choi Tim Antony Yung
* Simulation with A to B open

.INCLUDE blackbox.cir

XBOX vth 0 BOX
.AC LIN 1 60 60
.PRINT AC v(vth)

.END
```

```
ECE2101L Lab7 Part B2: Measuring OUTPUT impedance of a black box with 130 Ohm resistor
* Author: Choi Tim Antony Yung
* Simulation using 130ohm resistor

.INCLUDE blackbox.cir

XBOX v0 0 BOX
R0 v0 0 130
.AC LIN 1 60 60
.PRINT AC v(v0)

.END

```

```
ECE2101L Lab7 Part B2: Measuring OUTPUT impedance of a black box with 620 Ohm resistor
* Author: Choi Tim Antony Yung
* Simulation using 620ohm resistor

.INCLUDE blackbox.cir

XBOX v0 0 BOX
R0 v0 0 620
.AC LIN 1 60 60
.PRINT AC v(v0)

.END
```